Equivalence of finite element solution using Rayleigh Ritz method and Galerkin method for finding axial deformation of a simple cantilever beam, and on how to determine if the approximate solution found is an exact solution.

by Nasser M. Abbasi
Student, Engineering Mechanics dept.
University of Wisconsin, Madison
oct 16, 2009

Abstract

Using a simple problem, it is shown that, using a global trial function, the Rayleigh Ritz variational method and the Galerkin method give the same solution. These solutions are compared to the exact solution. It is found that one can determine that the exact solution was reached by increasing the order of the trial function polynomial until the solution returned by Rayleigh Ritz or Galerkin method no longer changes.

The problem

A cantilever beam is loaded with traction load of q = c x (for some constant c). It is fixed on the left side. We need to find axial deformation u (x)



Solving using variational Ritz method

Set up the potential energy π_p and Minimize it

$$\pi_p = \int_0^L \frac{1}{2} A E \left(\frac{du}{dx}\right)^2 dx - \int_0^L q\, u(x)\, dx$$

where u (x) is the trial function . The trial function needs to satisfy only the essential conditions, which in this problem is given as u (0) = 0, so we start by assuming a trial function $u(x)$.

$$u(x) = a_0 + a_1 x + a_2 x^2$$

Since $u(0) = 0$, we find that $a_0 = 0$ hence

$$u = a_1 x + a_2 x^2$$

This trial function has 2 degrees of freedom they are called the generalized degrees of freedom. The trial function could have been selected to have only one degree of freedom i.e. $u(x) = a_1 x$ and this would also have worked but 2 degrees of freedom will give more accurate solution.

Start by defining the trial function

```
Remove["Global`*"]
u = Sum[a_i x^i, {i, 0, 2}]
```

$$a_0 + x\, a_1 + x^2\, a_2$$

Make the trial function satisfy essential boundary condition

```
sol = First@ Solve[(u /. x → 0) == 0, a_0]
```

$$\{a_0 \to 0\}$$

```
u = u /. sol
```

$$x\, a_1 + x^2\, a_2$$

Set up the potential energy functional

```
Π = ∫_0^L 1/2 A E (∂_x u)^2 dx - ∫_0^L c x u dx
```

$$-\frac{1}{3} c L^3 a_1 + \frac{1}{2} A L E a_1^2 - \frac{1}{4} c L^4 a_2 + A L^2 E a_1 a_2 + \frac{2}{3} A L^3 E a_2^2$$

Solve for $\frac{d\Pi}{da_i} = 0$ for $i = 1, 2$ in order to solve for a_1 and a_2

```
eq1 = ∂_a1 Π == 0
```

$$-\frac{c L^3}{3} + A L E a_1 + A L^2 E a_2 == 0$$

```
eq2 = ∂_a2 Π == 0
```

$$-\frac{c L^4}{4} + A L^2 E a_1 + \frac{4}{3} A L^3 E a_2 == 0$$

```
sol = First @Solve[{eq1, eq2}, {a_1, a_2}]
```

$$\left\{a_1 \to \frac{7 c L^2}{12 A E},\ a_2 \to -\frac{c L}{4 A E}\right\}$$

Now that we found a_1 and a_2 we have found our solution

```
solutionByRitzMethod = u /. %
```

$$\frac{7\,c\,L^2\,x}{12\,A\,E} - \frac{c\,L\,x^2}{4\,A\,E}$$

Solving the problem using Galerkin method

In this method, we start with the differential equation itself and integrate it with a test function we call $g(x)$. The differential equation for axial deformation with traction load $q(x)$ is $A\,E\,\frac{d^2u}{dx^2} = -q(x)$, hence our starting point is to solve the integral equation

$$I = \int_0^L \left(\mathrm{AE}\,\frac{d^2u}{dx^2} + q \right) g(x)\,dx = 0$$

We start by assuming a trial function $u = a_0 + a_1 x + a_2 x^2$. Now due to $u(0) = 0$ we find that $a_0 = 0$, hence $u = a_1 x + a_2 x^2$ Next, we do integration by parts the above integral and obtain

$$\mathrm{I} = \left[g(x)\,\mathrm{AE}\left(\frac{du}{dx}\right)\right]_0^L - \int_0^L \left(\mathrm{AE}\,\frac{du}{dx} + q \right) \frac{dg(x)}{dx}\,dx + \int_0^L q\,g(x)\,dx$$

But

$$\left[g(x)\,\mathrm{AE}\left(\frac{du}{dx}\right)\right]_0^L = g(L)\,\mathrm{AE}\,\frac{du(L)}{dx} - g(0)\,\mathrm{AE}\,\frac{du(0)}{dx}$$

and now since a test function must also satisfy the essential conditions, hence $g(0) = 0$ and also the trial function satisfies the natural boundary conditions at $x = L$ which is that there is no load at that point, hence this means $A\,E\,(du/dx) = 0$ since this is the load at the end. Hence the integral becomes

$$\mathrm{I} = -\int_0^L \left(\mathrm{AE}\,\frac{du}{dx} + q \right) \frac{dg(x)}{dx}\,dx + \int_0^L q\,g(x)\,dx = 0$$

We now pick the test function g(x). In Galerkin method, $g_i(x) = \frac{du}{da_i}$ hence $g_1(x) = \frac{du}{da_1} = x$ and $g_2(x) = \frac{du}{da_2} = x^2$ so now we set up the 2 equations and solve for a_1 and a_2

```
u = x a1 + x^2 a2;
g1 = x;
g2 = x^2;
q = c x;
```

```
eq1 = ∫_0^L (-A E ∂_x u ∂_x g_1 + q g_1) dx
```

$$\frac{c\,L^3}{3} - A\,L\,E\,a_1 - A\,L^2\,E\,a_2$$

```
eq2 = ∫_0^L (-A E ∂_x u ∂_x g_2 + q g_2) dx
```

$$\frac{c\,L^4}{4} - A\,L^2\,E\,a_1 - \frac{4}{3}\,A\,L^3\,E\,a_2$$

Now solve for a_1 and a_2

```
sol = First @Solve[{eq1 == 0, eq2 == 0}, {a_1, a_2}]
```

$$\left\{a_1 \to \frac{7\,c\,L^2}{12\,A\,E},\ a_2 \to -\frac{c\,L}{4\,A\,E}\right\}$$

Substitute the above into the original $u(x)$

```
solutionByGalerkinMethod = u /. sol
```

$$\frac{7\,c\,L^2\,x}{12\,A\,E} - \frac{c\,L\,x^2}{4\,A\,E}$$

We see that this solution is the same as the one found from the Ritz method

```
solutionByGalerkinMethod - solutionByRitzMethod
```

```
0
```

Exact solution for the problem

The ODE is

$$A\,E\,u''[x] = -q$$

With the essential conditions $u(0) = 0$ and with natural boundary condition $A\,E\,u'[x] = 0$ at $x = L$. Find exact solution:

```
Remove[u]
q = c x;
sol = First@DSolve[{A E u''[x] == -q, u[0] == 0, A E u'[L] == 0}, u[x], x];
solutionExact = u[x] /. sol;
Collect[solutionExact, x]
```

$$\frac{c\,L^2\,x}{2\,A\,E} - \frac{c\,x^3}{6\,A\,E}$$

Compare the above solution to the one found above, we see that the approximate solution is

$$u = \frac{7\,c\,L^2\,x}{12\,A\,E} - \frac{c\,L\,x^2}{4\,A\,E}$$

while the exact solution is

$$u_{exact=} \frac{c\,L^2\,x}{2\,A\,E} - \frac{c\,x^3}{6\,A\,E}$$

here is a plot of both solutions. Assume $A = 1$, $c = 1$, $E = 1$ and $L = 1$, we obtain

```
Needs["PlotLegends`"]
```

```
c = 1; E = 1; L = 1; A = 1;
Plot[{solutionExact, solutionByRitzMethod},
 {x, 0, L}, PlotStyle → {Red, Blue},
 PlotLegend → {"exact", "FEM"}, LegendPosition → {-2, -.4}]
```



How can we improve the FEM solution, can we make it match the exact solution?

Make the trial function a polynomial of higher order. Try

$$u = a_0 + a_1\,x + a_2\,x^2 + a_3\,x^3$$

and repeat Ritz method. Instead of making the same steps as above, let us write a function which accepts a trial function $u(x)$ and returns back the solution. Find define the function

```
findSolutionByRitzMethod[u_, x_] := Module[{eqs, coeffVar, Π},
  Clear[A, E, c, L];
  coeffVar = Flatten[Normal[CoefficientArrays[u, x]]];
  coeffVar = Select[coeffVar, UnsameQ[#, 0] &];
  Π = ∫_0^L 1/2 A E (D[u, x])^2 dx - ∫_0^L c x u dx;
  eqs = Table[ D[Π, coeffVar[[i]] ] == 0, {i, 1, Length[coeffVar]}];
  First @Solve[eqs, coeffVar]
 ]
```

Now we can call the above function for the trial function shown above to dtermine the values of the a_i's

```
trialFunction = a1 x + a2 x^2 + a3 x^3;
solutionByRitzMethodMoreTerms =
 findSolutionByRitzMethod[trialFunction, x]
```

$$\left\{a_1 \to \frac{c\, L^2}{2\, A\, E},\ a_2 \to 0,\ a_3 \to -\frac{c}{6\, A\, E}\right\}$$

```
solutionByRitzMethodMoreTerms = trialFunction /. solutionByRitzMethodMoreTerms
```

$$\frac{c\, L^2\, x}{2\, A\, E} - \frac{c\, x^3}{6\, A\, E}$$

We see now that the FEM solution is the same as the exact solution. We can confirm this by taking the difference between them

```
solutionByRitzMethodMoreTerms - solutionExact // Simplify
```

```
0
```

What happens if we add more terms to the trial functions? We should expect that the solution would not change. This is confirmed by the following table generated using the above function

```
numberOfTrialFunctions = 5;
(trialFunction = Table[Sum[ai x^i, {i, 1, j}], {j, 1, numberOfTrialFunctions}]) // TableForm
```

$x\, a_1$
$x\, a_1 + x^2\, a_2$
$x\, a_1 + x^2\, a_2 + x^3\, a_3$
$x\, a_1 + x^2\, a_2 + x^3\, a_3 + x^4\, a_4$
$x\, a_1 + x^2\, a_2 + x^3\, a_3 + x^4\, a_4 + x^5\, a_5$

Now find the solution using each of the above trial functions using Ritz method

```
tbl = Table[findSolutionByRitzMethod[trialFunction[[i]], x], {i, 1, numberOfTrialFunctions}];
tbl =
  Table[{trialFunction[[i]], trialFunction[[i]] /. tbl[[i]]}, {i, 1, numberOfTrialFunctions}];

Grid[{ {"TrialFunction", "Solution"}, Sequence @@ tbl}, Frame → All]
```

TrialFunction	Solution
$x\, a_1$	$\frac{c\, L^2\, x}{3\, A\, E}$
$x\, a_1 + x^2\, a_2$	$\frac{7\, c\, L^2\, x}{12\, A\, E} - \frac{c\, L\, x^2}{4\, A\, E}$
$x\, a_1 + x^2\, a_2 + x^3\, a_3$	$\frac{c\, L^2\, x}{2\, A\, E} - \frac{c\, x^3}{6\, A\, E}$
$x\, a_1 + x^2\, a_2 + x^3\, a_3 + x^4\, a_4$	$\frac{c\, L^2\, x}{2\, A\, E} - \frac{c\, x^3}{6\, A\, E}$
$x\, a_1 + x^2\, a_2 + x^3\, a_3 + x^4\, a_4 + x^5\, a_5$	$\frac{c\, L^2\, x}{2\, A\, E} - \frac{c\, x^3}{6\, A\, E}$

We see from the above that as more terms added to the trial function, after the cubic polynomial, the solution no longer changes. This implies, at least in this example, that the exact solution was reached. (Need to find a formal proof of this).